
Abstract
In French engineering curriculums, teachers are requested to initiate students into research. The
author has developed such an activity in the context of a course in acoustic modeling. Although
doing genuine research with students is unrealistic, the paper shows that it remains possible to
mimic a research activity at the borderline of mathematical modeling and scientific computing, by
resorting to symbolic manipulations, which enable innovating aspects.

Introduction
In French engineering curriculums, teachers are prompted to initiate students into research. Most of them make their career in industry so
the purpose is not to train future researchers but to make them aware of innovation and able to interact with researchers and to participate
into cooperations between industry and laboratories.
Although doing genuine research with students is unrealistic, it remains possible to mimic a research activity at the borderline of mathematical modeling and scientific computing, by resorting to computer algebra and symbolic programming. Indeed, because of the persistent
abuse of numerical techniques, symbolic manipulations remain underused and this topic is not very documented yet, still missing from
most textbooks; this allows to tackle classical questions from an original viewpoint.
The paper describes an educational experiment developed in the context of a course in acoustic modeling, based on distribution theory,
Green functions and the Kirchhoff integral for the mathematical aspects and computer algebra and symbolic programming for the computational ones. Like symbolic manipulations, the modeling approach, at the common borderline of mathematics, physics and computing, is
not very documented, especially in textbooks, mostly specialized by discipline. This enables innovating aspects and above all avoids
plagiarism with « ready-made Internet reports ».
A number of subjects are proposed to students, about such various topics as: active control, forced waves in guides, transfer or propagation
matrices, eigenfrequencies and eigenmodes of various structures, plane waves in elastic solids, viscoelastic damping… Working teams of
two or three students must choose different subjects. After a brief description of the context of the experiment, these are presented below,
by focusing on the computational aspects, so the paper should be thought of as excerpts of the experiment.

Context of the experiment
The context is an optional course in acoustic modeling, in a French « engineering school » (more or less a master of science in engineering), more precisely, a mathematical and computational supplement, turned onto symbolic manipulations; each year, up to thirty students
are concerned.

Mathematical support: Green functions and the Kirchhoff integral
The scientific frame of these investigations is a chapter devoted to Green functions [A, chapter B], based itself on distribution theory and
the convolution product. More precisely, the Green functions of the one dimensional Helmholtz (or d’Alembert) equation are at the heart
of most proposed subjects. Indeed, these are at the heart of vibration or wave propagation modeling. In a preliminary course, students are
taught to compute them by a classical procedure (replace x by |x| in the solution of the associated homogeneous equation), which yields
either:
G(x) =

sin(k x )
2k

or : G(x) = ±

, k=

e±i k x
2ik

ω
c

, k=

ω
c

The sine form is well suited for the description of stationary waves in finite length guides while the exponential form is better suited for the
modeling of centrifugal or centripetal waves (the so called delayed and advanced Green functions) in the infinite space. Moreover, the
solutions of the homogeneous Helmholtz equation put forward the decomposition into forward and backward waves, useful in particular
for the modeling in terms of acoustic impedances or transmission matrices.
ω
H(x) = h e±i k x , k = , h any coefficient
c

In the case of finite length guides (or sections of guides), the Kirchhoff integral (which in the one dimensional case amounts to values at
the boundary of an interval) yields matrix or fourpole models (see farther).
P(x) = p(a) G ' (x -− a) + p ' (a) G(x -− a) -− p(b) G ' (x -− b) -− p ' (b) G(x -− b)
where G is a Green function and P is the distribution with bounded support [a, b] associated with the initial field p, assumed to satisfy the
homogeneous Helmholz equation over the interval (interval without sources). The formula typically applies to the case when the boundary
conditions are the sources (the Melde string).

Computational support: Mathematica
While simulation has long been restricted to numerical computations, computer algebra and symbolic programming enable the manipulation of exact or approximate formulas with parameters, which facilitates reasonings about models and their solutions. In particular, thanks
to its rule based language, Mathematica enables the direct programming of mathematical formulas by means of patterns and transformation
rules, in the frame of a uniform syntax with a mathematical flavor [B].
Students are reminded of the basic principle of programming: programs are functions or procedures that apply to typed data structures (or
methods associated with classes of objects). They are incited to tackle the problems in terms of specifications and to specify the overall
assumptions, the input data and their types, the output data and their types.

Proposed subjects
The proposed subjects are included in the handout (in French) available at the following URL, this learning support being actually a
mathematical annex to [C].
https : /∕/∕ macmaths.ens2m.fr /∕ documents /∕ Green.pdf
The corresponding answer handout is available from the author (remi.barrere@ens2m.fr). Below, the theoretical justifications (proofs or
calculations) are just briefly recalled; these can be found in the document. The paper rather emphasizes their computational counterpart.
For educational and practical reasons, most references of the handout are in French. Here, we refer to the compilation [A] as an English
equivalent.

Preliminaries
In most cases, it turned out to be preferable to replace Abs by its piecewise counterpart, so the expressions containing it can be derived or
expanded if need be. The definition is supplemented by a couple of preliminary settings.
abs[x_] := Piecewise[{{-− x, x < 0}, {x, x ≥ 0}}]
SetOptions[Plot, ImageSize → 200];
SetOptions[Plot3D, ImageSize → 200];
Round[QuantityMagnitude[UnitConvert[Quantity[1, "SpeedOfSound"],
"Meters/∕Seconds"]]]
343
WolframAlpha["Frequency of A", {{"StandardFrequency", 1}, "NumberData"}]
(*⋆ Internet connection required *⋆)
440
In the whole paper, c is the velocity of waves, ω the angular frequency, x an abscissa and t the time. In some cases, polymorphism is used,
i.e., the same function applies to different configurations, in particular different types of arguments.

Active control
By superposing cleverly parametrized sources, a Green function of the Helmholtz equation allows to develop models of active control, at
least simple models valid in the case of harmonic waves with given frequency.
The Green function of the Helmholtz equation used below represents the centrifugal response to a source spatially localized at the origin
and varying sinusoidally in time. Its reciprocity property, checked just below, actually derives from the evenness of the function. The
graphic representation shows the field of complex amplitudes (either real and imaginary parts or module and argument) while the animation visualizes the waves.
ExpGreenHelmholtz1D[c_][x_, ω_] :=
-− c Exp[-− ⅈ ω abs[x] /∕ c] /∕ (2 ⅈ ω)

FullSimplify[PiecewiseExpand[
ExpGreenHelmholtz1D[c][x -− y, ω] -−
ExpGreenHelmholtz1D[c][y -− x, ω]],
x -− y > 0 || x -− y < 0 || x -− y ⩵ 0]
0
GH1D[x_] := ExpGreenHelmholtz1D[343][x, 2 π 440]
GH1D[x_, t_] := ExpGreenHelmholtz1D[343][x, 2 π 440] Exp[ⅈ 2 π 440 t]
GraphicsRow[{
Plot[Evaluate[{Re[GH1D[x]], Im[GH1D[x]]}], {x, -− 1.5, 1.5}],
Plot[Evaluate[{Log10[Abs[GH1D[x]]], Arg[GH1D[x]]}], {x, -− 1.5, 1.5},
Exclusions → None, PlotRange → {-− 3.5, 3.5}]
}]

Animate[Plot[Evaluate[{Re[GH1D[x, t]], Im[GH1D[x, t]]}], {x, -− 1.5, 1.5}],
{t, 0, Infinity}, AnimationRate → 0.001, AnimationRunning → False]

Let us introduce a second point harmonic source at abscissa d, with amplitude a and phase offset φ. Let us superpose it with the former,
then compute and visualize the effect. The condition to cancel the field over the domain x > d is: a = 1  ω d /∕ c + φ + π = 0 [modulo 2 π].
ActiveControl[c_, d_, a_, φ_][x_, ω_] := a ExpGreenHelmholtz1D[c][x -− d, ω] Exp[ⅈ φ]
ControlledField[c_, d_, a_, φ_][x_, ω_] :=
ExpGreenHelmholtz1D[c][x, ω] + ActiveControl[c, d, a, φ][x, ω]
CF[d_, φ_][x_] := ControlledField[343, d, 1, φ][x, 2 π 440]
CFComponents[d_, φ_][x_] := {Re[CF[d, φ][x]], Im[CF[d, φ][x]], Abs[CF[d, φ][x]]}
Here are the graphic representations of two typical cases when the condition is true.

GraphicsRow[{
Plot[Evaluate[With[{d = π 343 /∕ (2 π 440)}, CFComponents[d, 0][x] ]],
{x, -− 0.5, 1.}, PlotRange → {-− 0.15, 0.15}],
Plot[Evaluate[With[{d = 343 (π /∕ 2) /∕ (2 π 440)}, CFComponents[d, π /∕ 2][x] ]],
{x, -− 0.5, 1.}, PlotRange → {-− 0.15, 0.15}]
}]

According to the pair {d, φ} that satisfies the condition, various configurations are possible, which can be explored with Manipulate.
Manipulate[Plot[Evaluate[With[{d = 343 (π -− φ) /∕ (2 π 440)},
CFComponents[d, φ][x]]], {x, -− 1.5, 1.5},
PlotRange → {-− 0.2, 0.2}], {{φ, 0}, -− 2 π, 3 π}]

This single frequency model gives an idea of the principle behind active control. More sophisticated techniques are used in practice (for
instance in airplanes), possibly including signal processing.

Acoustic impedances
From the solution of the homogeneous Helmholtz equation, one recovers the formulation in terms of acoustic impedances (this section) or
else transmission matrices (further section).
The input impedance formula gives the equivalent impedance of any terminating section of a waveguide. Zs = ρ c is the specific
impedance, Zt the terminal impedance, y its abscissa and x the abscissa at which the impedance is computed.
InputImpedance[c_, Zs_, Zt_, y_][x_, ω_] :=
Zs (Zt + ⅈ Zs Tan[ω (y -− x) /∕ c]) /∕ (Zs + ⅈ Zt Tan[ω (y -− x) /∕ c])
For instance, if Zs is real, an open duct of any length l has an imaginary impedance, hence is a means to make up an imaginary impedance.

InputImpedance[c, Zs, 0, x + l][x, ω]
ⅈ Zs Tan

lω

c

Let us l vary from 0 to λ/2, with λ the wavelength. Then the coefficient l/λ varies from 0 to 1/2.
Tan[ω l /∕ c] /∕. ω → 2 π c /∕ λ
Tan

2lπ

λ

Plot[Tan[2 loverλ π], {loverλ, 0, 1 /∕ 2}]

By adjusting the length of the duct, any value of the imaginary part of the impedance in ] -− ∞, +∞[ can be achieved (the factor Zs must
nonetheless be taken into account in the computation), with a singularity at l = λ /∕ 4, where the impedance becomes infinite.
The input impedance formula is classically used to model multisection systems, especially multilayered materials. To carry out a program,
we must firstly find the change of impedance through a diopter, i.e., an interface between sections with different characteristics (section
areas, S1 and S2 below, or specific impedances, Z1 and Z2 below). Actually, the expression of the continuity of pressures and volume
flows at the interface leads to the formula:
Z1
S1

=

Z2
S2

Hence the program. The specific impedances Z1 and Z2 are introduced in Diopter[{S1, Z1}, {S2, Z2}] in view of further use for transmission matrices.
InputImpedance[Duct[l_, c_, Zs_]][Zt_, ω_] :=
Zs (Zt + ⅈ Zs Tan[ω l /∕ c]) /∕ (Zs + ⅈ Zt Tan[ω l /∕ c])
InputImpedance[Diopter[{S1_, _}, {S2_, _}]][Zt_, ω_] := Zt *⋆ S1 /∕ S2
InputImpedance[Cascade[w_]][Zt_, ω_] :=
Together[Fold[InputImpedance[#2][#1, ω] &, Zt, Reverse[w]]]
Zi = InputImpedance[Cascade[{Diopter[{S0, Z0}, {S, Zs}], Duct[l, c, Zs]}]][Zt, ω]
S0 Zs Zt + ⅈ Zs Tan
S Zs + ⅈ Zt Tan

lω

c

lω

c

Taking into account the redundancy in the data structure, it might be preferred to input only the list of ducts and let the program compute
the diopters.
InputImpedance[Duct[l_, c_, Zs_, _]][Zt_, ω_] :=
Zs (Zt + ⅈ Zs Tan[ω l /∕ c]) /∕ (Zs + ⅈ Zt Tan[ω l /∕ c])
BuildDiopter[{Duct[_, _, Z1_, S1_], Duct[_, _, Z2_, S2_]}] :=
Diopter[{S1, Z1}, {S2, Z2}]
InputImpedance[Chain[w_]][Zt_, ω_] :=
InputImpedance[
Cascade[DeleteCases[Riffle[w, Map[BuildDiopter, Partition[w, 2, 1]]],
Duct[0, __]]]][Zt, ω]

InputImpedance[Chain[{
Duct[0, c, Z0, S0], Duct[l, c, Zs, S]}]][Zt, ω] ⩵ Zi
True
This may require to introduce a zero length duct to get the impedance just before a diopter. The program can be used to investigate the
frequency response of multisection systems, typically multilayered materials.

Transmission matrices
In continuation of the impedance approach, transmission matrices are used for the modeling of assemblies of waveguide sections. Firstly,
we focus on what happens through a diopter, i.e., an interface between sections with different characteristics (section areas, S1 and S2
below, or specific impedances, Z1 and Z2 below).
The expression of the continuity of pressures and volume flows at the interface leads to a system of equations between the complex
amplitudes of the forward and backward waves on either side: pfm (forward minus), pbm (backward minus), pfp (forward plus), pbp
(backward plus).
Solve[{pfm + pbm ⩵ pfp + pbp, S1 (pfm -− pbm) /∕ Z1 ⩵ S2 (pfp -− pbp) /∕ Z2}, {pfp, pbp}]
-− pbm S2 Z1 -− pfm S2 Z1 + pbm S1 Z2 -− pfm S1 Z2
pfp → -−
pbp → -−

2 S2 Z1
-− pbm S2 Z1 -− pfm S2 Z1 -− pbm S1 Z2 + pfm S1 Z2

,


2 S2 Z1

Solving the system with respect to pfp and pbp yields the transmission matrix while its solution with respect to pbm and pfm yields its
inverse.
TransmissionMatrix[Diopter[{S1_, Z1_}, {S2_, Z2_}], _ : ω] =
Module[{pfm, pbm, pfp, pbp},
Together[
Normal[CoefficientArrays[
{pfp, pbp} /∕. Solve[{pfm + pbm ⩵ pfp + pbp, S1 (pfm -− pbm) /∕ Z1 ⩵ S2 (pfp -− pbp) /∕ Z2},
{pfp, pbp}], {pfm, pbm}]]][[2, 1]]
]
S2 Z1 + S1 Z2

2 S2 Z1

,

S2 Z1 -− S1 Z2

S2 Z1 -− S1 Z2
, 

2 S2 Z1

2 S2 Z1

,

S2 Z1 + S1 Z2

2 S2 Z1

InverseTransmissionMatrix[Diopter[{S1_, Z1_}, {S2_, Z2_}], _ : ω] =
Module[{pfm, pbm, pfp, pbp},
Together[
Normal[CoefficientArrays[
{pfm, pbm} /∕. Solve[{pfm + pbm ⩵ pfp + pbp, S1 (pfm -− pbm) /∕ Z1 ⩵ S2 (pfp -− pbp) /∕ Z2},
{pfm, pbm}], {pfp, pbp}]]][[2, 1]]
]
S2 Z1 + S1 Z2

2 S1 Z2

,

-− S2 Z1 + S1 Z2

-− S2 Z1 + S1 Z2
, 

2 S1 Z2

2 S1 Z2

,

S2 Z1 + S1 Z2

2 S1 Z2

Simplify[%.%%]
{{1, 0}, {0, 1}}
When pbp equals zero (infinite guide or adapted terminal impedance), the matrix relation gives the reflection and transmission coefficients,
defined by T = p f p  p f m and R = pb m  p f m , i.e., {T, 0} = TR.{1, R} with TR the transmission matrix.
Solve[Thread[{T, 0} ⩵ {{t11, t12}, {t21, t22}}.{1, R}], {R, T}]
t21
R → -−
t22

, T → -−

t12 t21 -− t11 t22

t22

ReflexionCoefficient[tr_ ? MatrixQ] := -− Apply[Divide, Last[tr]]
TransmissionCoefficient[tr_ ? MatrixQ] := Det[tr] /∕ tr[[2, 2]]

This yields the classical formulas for the reflexion and transmission coefficients in the case of a diopter; but any device, considered as a
whole, has a reflection and a transmission coefficient.
TR = TransmissionMatrix[Diopter[{S1, Z1}, {S2, Z2}]];
ReflexionCoefficient[TR]
S2 Z1 -− S1 Z2
-−
S2 Z1 + S1 Z2
TransmissionCoefficient[TR]
2 S1 Z2
S2 Z1 + S1 Z2
Resorting to the hybrid pairs p f m , pb p  and pb m , p f p  leads to the so called S matrix (or its inverse), a generalization of which is used to
model guide junctions. The expression of the pressure along a duct section (general solution of the homogeneous Helmholtz equation):
-− i

p(x, ω) = p f e

ω
c

x

+i

+ pb e

ω
c

x

leads to its transmission matrix. Finally, the transmission matrix of a chain of devices is the product of their transmission matrices.
TransmissionMatrix[Duct[l_, c_], ω_] := {{Exp[ⅈ ω l /∕ c], 0}, {0, Exp[-− ⅈ ω l /∕ c]}}
InverseTransmissionMatrix[Duct[l_, c_], ω_] :=
{{Exp[-− ⅈ ω l /∕ c], 0}, {0, Exp[ⅈ ω l /∕ c]}}
TransmissionMatrix[Cascade[w_], ω_] :=
Apply[Dot, Map[TransmissionMatrix[#, ω] &, Reverse[w]]]
InverseTransmissionMatrix[Cascade[w_], ω_] :=
Apply[Dot, Map[InverseTransmissionMatrix[#, ω] &, w]]
ReflexionAndTransmissionCoefficients[device_, ω_] :=
Through[{ReflexionCoefficient, TransmissionCoefficient}[
TransmissionMatrix[device, ω]]]
tm = TransmissionMatrix[setting = Cascade[{
Diopter[{Sm, Z}, {S, Z}],
Duct[l, c],
Diopter[{S, Z}, {S, Zp}]
}], ω]
ⅇ

-−

ⅈ l ω

ⅈ l ω

(S Z -− Sm Z) (S Z -− S Zp)

c



2

4S Z
ⅇ

-−

ⅇ

4S Z

ⅇ

(S Z + Sm Z) (S Z -− S Zp)



(S Z -− Sm Z) (S Z + S Zp)

ⅇ

-−

ⅈ l ω
c

(S Z -− Sm Z) (S Z + S Zp)
4 S2 Z2

ⅈ l ω

(S Z -− Sm Z) (S Z -− S Zp)

ⅇ

-−

,

ⅈ l ω
c

(S Z + Sm Z) (S Z + S Zp)

+

4 S2 Z2

,

4 S2 Z2

+

4 S2 Z2
c

c

+

2

ⅈ l ω

ⅇ

,

ⅈ l ω

(S Z + Sm Z) (S Z -− S Zp)

c

c

(S Z + Sm Z) (S Z + S Zp)
4 S2 Z2

ⅈ l ω

2

ⅇ

c

+

2



4 S2 Z2

ReflexionAndTransmissionCoefficients[setting, ω] /∕/∕ Together
2 ⅈ l ω

S Z + ⅇ
-−

2 ⅈ l ω

S Z -− Sm Z + ⅇ

c
2 ⅈ l ω

S Z + ⅇ

c

2 ⅈ l ω

Sm Z + S Zp -− ⅇ

c
2 ⅈ l ω

S Z + Sm Z -− ⅇ

c

2 ⅈ l ω

S Zp -− Sm Zp -− ⅇ

c
2 ⅈ l ω

Sm Z + S Zp -− ⅇ

c

Sm Zp

c
2 ⅈ l ω

S Zp + Sm Zp + ⅇ

c

Sm Zp

ⅈ l ω

4 ⅇ
2 ⅈ l ω

S Z + ⅇ

c

c

Sm Zp

2 ⅈ l ω

S Z + Sm Z -− ⅇ

c

2 ⅈ l ω

Sm Z + S Zp -− ⅇ

These can then be investigated with respect to frequency. Taking into account the redundancy in the data structure, it might be preferable
to input the list of ducts and let the program compute the diopters.
TransmissionMatrix[Duct[l_, c_, Zc_, S_], ω_] :=
{{Exp[ⅈ ω l /∕ c], 0}, {0, Exp[-− ⅈ ω l /∕ c]}}
BuildDiopter[{Duct[_, _, Z1_, S1_], Duct[_, _, Z2_, S2_]}] :=
Diopter[{S1, Z1}, {S2, Z2}]
TransmissionMatrix[Chain[w_], ω_] :=
TransmissionMatrix[
Cascade[DeleteCases[Riffle[w, Map[BuildDiopter, Partition[w, 2, 1]]],
Duct[0, _, _, _]]], ω]
TransmissionMatrix[Chain[{
Duct[0, c, Z, Sm], Duct[l, c, Z, S], Duct[0, c, Zp, S]
}], ω] ⩵ tm
True
This technique based on the transmission matrix is used for the modeling of multilayered systems (acoustic filters, walls, double glazings,
compound absorbers…).

Forced vibrations
As an alternative to the classical superposition of eigenmodes, the sine Green function of the Helmholtz equation over an interval represents the response to a point harmonic source, which can then be generalized to any source by Fourier superposition. The program below is
parametrized by the boundary conditions (Dirichlet or Neumann).
Let us begin with the general form of the Green function of the infinite space, obtained by superposing the general solution of the associated homogeneous equation.
SineGreenHelmholtz1D[c_, d_, c1_: c1, c2_: c2][x_, ω_] :=
c *⋆ Sin[ω abs[x -− d] /∕ c] /∕ (2 ω) + c1 Cos[ω x /∕ c] + c2 Sin[ω x /∕ c]
The boundary conditions yield a 2 × 2 linear system that determines the integration constants c1 and c2 . The program is designed so as to
directly compute the matrices and then use LinearSolve.
Boundary[c_, d_, ω_, HomogeneousDirichlet[x_]] :=
{{ Cos[ω x /∕ c], Sin[ω x /∕ c]}, -− c *⋆ Sin[ω abs[x -− d] /∕ c] /∕ (2 ω)}
Boundary[c_, d_, ω_, HomogeneousNeumann[x_]] :=
{{ -− ω Sin[ω x /∕ c] /∕ c, ω Cos[ω x /∕ c] /∕ c},
-− Derivative[1][c *⋆ Sin[ω abs[# -− d] /∕ c] /∕ (2 ω) &][x]}
IntervalGreenHelmholtz1D[c_, d_, lbc_, rbc_][x_, ω_] :=
Module[{c1, c2}, SineGreenHelmholtz1D[c, d, c1, c2][x, ω] /∕.
Thread[
Rule[{c1, c2}, Apply[LinearSolve,
Transpose[{Boundary[c, d, ω, lbc], Boundary[c, d, ω, rbc]}]]]] ]
The program applies to any pair of boundary conditions (Dirichlet or Neuman) and yields a quite simple closed form solution. A complementary transformation rule would be necessary for Robin conditions.
IGHDN[c_, d_][x_, ω_] = FullSimplify[IntervalGreenHelmholtz1D[c, d,
HomogeneousDirichlet[0], HomogeneousNeumann[1]
][x, ω], 0 < d < 1]
c Cos

-−

c

ω

xω

c

c

ω

dω

c

c

 Sec  Sin



ω
c Cos

-−

ω-−d ω

ω-−x ω
c

 Sec  Sin



ω

d>x
True

The resonance angular frequencies (eigenfrequencies of the system) can be recovered by canceling the denominator. From the general
formula below, the first five frequencies are computed.
Solve[Cos[ω /∕ c] ⩵ 0, ω]
ω → ConditionalExpressionc -−
ω → ConditionalExpressionc

π
2

π
2

+ 2 π C[1] , C[1] ∈ Integers,

+ 2 π C[1] , C[1] ∈ Integers

N[Function[ 343 (π /∕ 2 + π #) /∕ (2 π)][Range[0, 4]]] (*⋆ Hertz *⋆)
{85.75, 257.25, 428.75, 600.25, 771.75}
The interactive pane shows the field of amplitudes parametrized by the frequency and the abscissa of the point source. The derivative is
discontinuous at x = d. For the resonance frequencies, the amplitudes become infinite and there is a phase leap.
Manipulate[Plot[Evaluate[{
IGHDN[343, d][x, 2 π ν], D[IGHDN[343, d][x, 2 π ν], x] /∕ 12}
], {x, 0, 1}, PlotRange → {-− 0.5, 0.5}, PlotStyle → {Opacity[1], Opacity[0.3]}],
{{d, 0.304}, 0, 1}, {{ν, 440}, 50, 900}]

d
ν

0.4

0.2

-−1.0

-−0.5

1.0

-−0.2

In the case of two Dirichlet conditions, we alternately give a two-dimensional representation.
IGHDD[c_, d_][x_, ω_] = FullSimplify[IntervalGreenHelmholtz1D[c, d,
HomogeneousDirichlet[0], HomogeneousDirichlet[1]
][x, ω], 0 < d < 1]
ω

c Csc  Sin

-−

c

 Sin

ω-−d ω
c



ω
ω

c Csc  Sin

-−

xω

c

dω

c

c

 Sin

ω-−x ω
c



ω

d>x
True

Solve[Sin[ω /∕ c] ⩵ 0, ω]
{{ω → ConditionalExpression[2 c π C[1], C[1] ∈ Integers]},
{ω → ConditionalExpression[c (π + 2 π C[1]), C[1] ∈ Integers]}}
N[Function[ 343 π # /∕ (2 π)][Range[4]]] (*⋆ Hertz *⋆)
{171.5, 343., 514.5, 686.}
The surface shows how (for a given value of d) the field of amplitudes evolves with the frequency. It must be understood that the amplitude becomes infinite for the resonance frequencies.

12th International Mathematica Symposium, 2015, Prague, Czech Republic

10 / 19

Plot3D[IGHDD[343, 0.304][x, 2 π ν], {x, 0, 1}, {ν, 0, 800},
PlotRange → {-− 0.3, 0.3}, PlotPoints → 35, MaxRecursion → 4, Exclusions → None,
ViewPoint → {2.15, -− 1., 2.35}, ImageSize → 300]
(*⋆ Please, be patient *⋆)

Finally, the following Manipulate analyses the response with respect to x an ω, when d varies from 0 to 1. In particular, a resonance
disappears when d is one of the nodes of the corresponding eigenmode. For instance, for d = 1 /∕ 2, the even eigenmodes vanish.
Manipulate[Plot3D[IGHDD[343, d][x, 2 π ν], {x, 0, 1}, {ν, 0, 800},
PlotRange → {-− 0.3, 0.3}, PlotPoints → 35, Exclusions → None,
ViewPoint → {2.15, -− 1., 2.35}, ImageSize → 250], {{d, 0.5}, 0, 1}]

d

0.0
0.2
0.0
-−0.2
0.5
800
600
400
200
1.0

0

Kirchhoff’s integral and the fourpole (matrix) approach
In the one-dimensional case, the Kirchhoff integral boils down to values at the boundary of an interval. It actually computes the distribution P with bounded support [a, b] associated with the initial field p, assumed to satisfy the homogeneous Helmholz equation, G being a
Green function.
P(x) = p(a) G ' (x -− a) + p ' (a) G(x -− a) -− p(b) G ' (x -− b) -− p ' (b) G(x -− b)
Let us compute P and verify it is zero outside the interval, for a given p and a given interval, say [0, 1].
KirchhoffBox1D[c_, {a_, {pa_, dpa_}}, {b_, {pb_, dpb_}}][x_, ω_] =
Convolve[pa DiracDelta '[y -− a] + dpa DiracDelta[y -− a] -− pb DiracDelta '[y -− b] -−
dpb DiracDelta[y -− b], Sin[ω abs[y] /∕ c] *⋆ c /∕ (2 ω), y, x];

example[x_] = (*⋆ ω = 1; c = 1 *⋆)
y[x] /∕. First[DSolve[{y ''[x] + y[x] ⩵ 0, y[0] ⩵ 1, y '[1] ⩵ 0}, y[x], x]]
Cos[x] + Sin[x] Tan[1]
boundaryValues = {example[0], example '[0], example[1], example '[1]}
{1, Tan[1], Cos[1] + Sin[1] Tan[1], 0}
Plot[Evaluate[KirchhoffBox1D[1, {0, {pa, dpa}}, {1, {pb, dpb}}][x, 1] /∕.
Thread[Rule[{pa, dpa, pb, dpb}, boundaryValues]]], {x, -− 0.5, 1.5}]

1.5
1.0
0.5

0.5

-−0.5

1.0

1.5

For simplicity in the following, we manipulate only the expression valid inside the interval.
Kirchhoff1D[c_, {a_, {pa_, dpa_}}, {b_, {pb_, dpb_}}][x_, ω_] =
Simplify[KirchhoffBox1D[c, {a, {pa, dpa}}, {b, {pb, dpb}}][x, ω], a < x < b]
pb ω Cos

(b-−x) ω

c

+ pa ω Cos

(-−a+x) ω

c

-− c dpb Sin

(b-−x) ω

c

+ c dpa Sin

(-−a+x) ω

c

2ω
Kirchhoff1D[c, {a, {pa, dpa}}, {b, {pb, dpb}}][a, ω]
pa ω + pb ω Cos

(-−a+b) ω

c

-− c dpb Sin

(-−a+b) ω

c

2ω
Derivative[1][Kirchhoff1D[c, {a, {pa, dpa}}, {b, {pb, dpb}}][#, ω] &][b]
dpb ω + dpa ω Cos

(-−a+b) ω

c

pa ω2 Sin

-−

(-−a+b) ω
c



c

2ω
By specifying the values at the boundaries of either p or p’, and by solving with respect to any pair chosen among {pa, dpa, pb, dpb}, an
expression that depends on the complementary pair is obtained; for instance:
transfer =
Simplify[
{pb, dpb} /∕.
First[Solve[{Kirchhoff1D[c, {a, {pa, dpa}}, {b, {pb, dpb}}][a, ω] ⩵ pa,
Kirchhoff1D[c, {a, {pa, dpa}}, {b, {pb, dpb}}][b, ω] ⩵ pb}, {pb, dpb}]]]
pa Cos

c dpa Sin

(-− a + b) ω
+
c

(-−a+b) ω

c

, dpa Cos

pa ω Sin

(-− a + b) ω

ω

 -−
c

(-−a+b) ω

c



c

This is the well known transfer relation between the input at a and the output at b of the corresponding section of the waveguide, which
depends only on b -− a. It is generally expressed in matrix terms, parametrized directly by the length of the duct.
TransferMatrix[Duct[l_, c_], ω_] =
Simplify[Normal[CoefficientArrays[transfer, {pa, dpa}][[2]]] /∕. {b -− a → l}]
c Sin

lω
Cos

,
c

ω

lω

c

ω Sin
, -−

lω

c

c

, Cos

lω

c

This is actually a mathematical version of the transfer matrix where the dual quantity is the derivative. In acoustics, the dual quantity is
rather the velocity (or sometimes the volume flow velocity), defined by the following property, with Zs the specific impedance and ρ the
mass density (Zs = ρ c):

This is actually a mathematical version of the transfer matrix where the dual quantity is the derivative. In acoustics, the dual quantity is
rather the velocity (or sometimes the volume flow velocity), defined by the following property, with Zs the specific impedance and ρ the
mass density (Zs = ρ c):
∂v
ρ

∂t

∂p
= -−

∂x

, hence : v =

-−1 ∂ p
i ω ρ ∂x
cos

Hence the transfer matrix :
-−

i
Zs

c
=

ωl

ω Zs ∂ x

-−i Zs sin

c

sin

i ∂p

ωl
c

cos

ωl
c

ωl
c

This leads to a fourpole approach to sections of waveguides. According to the nature of the initial problem, any other pair chosen among
{pa, dpa, pb, dpb} can be computed from the complementary pair, thus leading to the corresponding (impedance, admittance, hybrid…)
matrices. In parallel, fourpole theory provides with linear relations between these different matrices, which can then be computed from one
of them.
For instance a sequence of duct sections is modeled by the product of transfer matrices. Any section of a waveguide can be viewed as a
fourpole and treated that way, thus involving methods that come from electronics. This is used for the modeling of acoustic filters,
mufflers, multilayered materials…

The Melde string
The Melde string is actually an emblem of the case when boundary conditions are sources. The original problem was a vibrating string
driven by a boundary condition but it applies as well to finite length waveguides or any system modeled by the d’Alembert equation or the
associated Helmholtz one (e.g., longitudinal or torsional vibrations in rods). Below, we tackle the problem firstly by the classical approach,
then by means of the Kirchhoff formula, which leads to impedance, admittance, hybrid or transfer matrices, from which the field over the
interval is recovered.
The Helmholtz equation operating in the Fourier space, an oscillating boundary condition is characterized by its amplitude, say p0 . The
procedure consists in determining the integration constants of the general solution of the homogeneous equation.
SineHelmholtz1D[c_, c1_: c1, c2_: c2][x_, ω_] := c1 Cos[ω x /∕ c] + c2 Sin[ω x /∕ c]
Boundary[c_, ω_, Dirichlet[x_, v_]] :=
{{ Cos[ω x /∕ c], Sin[ω x /∕ c]}, v}
Boundary[c_, ω_, Neumann[x_, v_]] := {{ -− ω Sin[ω x /∕ c] /∕ c, ω Cos[ω x /∕ c] /∕ c}, v}
The program builds up the linear system that expresses the boundary conditions, then solves it and substitutes the general solution with the
computed constants.
IntervalHelmholtz1D[c_, lbc_, rbc_][x_, ω_] :=
Module[{c1, c2}, SineHelmholtz1D[c, c1, c2][x, ω] /∕.
Thread[
Rule[{c1, c2}, Apply[LinearSolve,
Transpose[{Boundary[c, ω, lbc], Boundary[c, ω, rbc]}]
] ]] ]
IHDD[c_, p0_][x_, ω_] = Simplify[IntervalHelmholtz1D[c,
Dirichlet[0, p0], Dirichlet[1, 0]
][x, ω]]
ω
ω -− x ω
p0 Csc  Sin

c
c
Solve[Sin[ω /∕ c] ⩵ 0, ω]
{{ω → ConditionalExpression[2 c π C[1], C[1] ∈ Integers]},
{ω → ConditionalExpression[c (π + 2 π C[1]), C[1] ∈ Integers]}}
N[Function[ 343 π # /∕ (2 π)][Range[4]]] (*⋆ Hertz *⋆)
{171.5, 343., 514.5, 686.}
The interactive pane shows the field of amplitudes parametrized by the frequency. For the resonance frequencies, the amplitudes become
infinite and there is a phase jump, as expected.

Manipulate[Plot[IHDD[343, 1][x, 2 π ν], {x, 0, 1},
PlotRange → {-− 5, 5}], {{ν, 440}, 50, 800}]

ν

4

2

-−1.0

0.5

-−0.5

1.0

-−2

-−4

The following dynamic pane simulates the stationary waves, that result from the interference between the forward and backward waves.
Manipulate[Plot[IHDD[343, 1][x, 2 π ν] Cos[2 π ν t], {x, 0, 1},
PlotRange → {-− 5, 5}], {{ν, 440}, 50, 800},
{t, 0, Infinity, ControlType → Animator, AnimationRate → 0.002,
AnimationRunning → False}]

ν
t

4

2

-−1.0

0.5

-−0.5

1.0

-−2

-−4

The two-dimensional representation shows how the field of amplitudes evolves with the frequency. As expected, the amplitude becomes
infinite for the resonance frequencies.

12th International Mathematica Symposium, 2015, Prague, Czech Republic

14 / 19

Plot3D[IHDD[343, 1][x, 2 π ν], {x, 0, 1}, {ν, 0, 800},
PlotRange → {-− 2.5, 2.5}, PlotPoints → 35, MaxRecursion → 4, Exclusions → None,
ViewPoint → {2.15, -− 1., 2.35}, ImageSize → 300]
(*⋆ Please, be patient *⋆)

Now, we show that this result can also be obtained by means of the Kirchhoff formula. The admittance introduced below is the mathematical one, corresponding to the pair {p, p '}; in acoustics, it is rather associated with the pair {p, v}, hence involves the specific impedance.
AdmittanceRules[c_, {a_, {pa_, dpa_}}, {b_, {pb_, dpb_}}][ω_] =
First[Solve[{Kirchhoff1D[c, {a, {pa, dpa}}, {b, {pb, dpb}}][a, ω] ⩵ pa,
Kirchhoff1D[c, {a, {pa, dpa}}, {b, {pb, dpb}}][b, ω] ⩵ pb}, {dpa, dpb}]]
ω pb -− pa Cos
dpa →

(-−a+b) ω

c

Csc

(-−a+b) ω

c

c

ω pa -− pb Cos
, dpb → -−

(-−a+b) ω

c

Csc

(-−a+b) ω

c



c

This is actually the admittance relation between the boundary values at a and b and their derivatives, which depends only on b -− a. It is
generally expressed in matrix terms, parametrized directly by the length of the duct, but here, we directly reuse the rules to get the field
over the interval.
MeldeHelmholtz1D[c_, Dirichlet[a_, va_], Dirichlet[b_, vb_]][x_, ω_] :=
Module[{pa, dpa, pb, dpb},
Simplify[Kirchhoff1D[c, {a, {pa, dpa}}, {b, {pb, dpb}}][x, ω] /∕.
AdmittanceRules[c, {a, {pa, dpa}}, {b, {pb, dpb}}][ω] /∕. {pa → va, pb → vb}]
]
MeldeHelmholtz1D[c, Dirichlet[0, p0 ], Dirichlet[1, 0]][x, ω]
ω
ω -− x ω
Csc  Sin
 p0
c
c
The former (classical) computation is straightforward while the latter one is more intricate. It nonetheless leads to an unusual interpretation
of the Kirchhoff formula in terms of fourpole matrices. It is actually a variant of that obtained at the ,previous section. The question
remains of its generalization to higher dimensions.

Propagation modes in anisotropic media
The analysis of plane wave propagation in anisotropic elastic solids leads to the general schema of 3 propagation modes with orthogonal
polarizations. This result can then be reused to show that an incident wave on a plane mechanical diopter (interface between two different
media) generates up to 3 reflected waves and up to 3 refracted waves [D].
In the absence of external body-forces, wave propagation in elastic media is governed by a set of 3 partial differential equations, with ui
the displacement, ρ the mass density and e i j k l the elastic tensor.
ρ

∂2 ui
∂ t2

= e i j k l
jkl

∂2 u k
∂ x j ∂ xl

Seeking harmonic plane waves with propagation direction v:

i ω t -−

ui = ai ei(ω t -− k v.x) = ai e

v.x
c



with : k =

ω
c

leads to an eigenvalue and eigenvector problem:
ρ c2 ai = e i j k l v j vl ak
jkl

Indeed, by introducing the Christoffel contracted tensor Γ i k = ∑ j l e i j k l v j vl , the system rewrites:
ρ c2 ai = ∑ k Γ i k ak
The propagation velocities derive from the eigenvalues while the associated eigenvectors give the polarizations (directions of the mechanical displacements). Because of the tensor symmetries, the eigenvalues are real and the eigenvectors orthogonal. That means there are 3
propagation modes, in general a quasi-longitudinal (compression) wave and two quasi-transversal (shear) wave, so a plane wave with
given frequency is a superposition of them.
In practice, the fourth-rank elastic tensor is often coded in matrix form (a 6 × 6 matrix), thanks to the Voigt transformation of indices:
Voigt[1, 1] = 1; Voigt[2, 2] = 2; Voigt[3, 3] = 3;
Voigt[2, 3] = 4; Voigt[1, 3] = 5; Voigt[1, 2] = 6;
Voigt[3, 2] = 4; Voigt[3, 1] = 5; Voigt[2, 1] = 6;
The number of independant coefficients depends on the crystal symmetries, so a compact coding is obtained by introducing types for the
symmetries, for instance:
StiffnessMatrix[Isotropic[{c44_, c12_}]] := SparseArray[{
{i_, i_} /∕; 1 ≤ i ≤ 3 ⧴ c12 + 2 c44, {i_, i_} /∕; 4 ≤ i ≤ 6 ⧴ c44,
{i_, j_} /∕; 1 ≤ i < j ≤ 3 ⧴ c12, {i_, j_} /∕; 1 ≤ j < i ≤ 3 ⧴ c12
}, {6, 6}]
StiffnessMatrix[Cubic[{c11_, c44_, c12_}]] := SparseArray[{
{i_, i_} /∕; 1 ≤ i ≤ 3 ⧴ c11, {i_, i_} /∕; 4 ≤ i ≤ 6 ⧴ c44,
{i_, j_} /∕; 1 ≤ i < j ≤ 3 ⧴ c12, {i_, j_} /∕; 1 ≤ j < i ≤ 3 ⧴ c12
}, {6, 6}](*⋆ and so forth for other symmetries *⋆)
A database can be introduced in the form of functions (defined by cases).
MassDensity["Silicon"] := 2332. (*⋆ SI *⋆)
StiffnessCoefficients["Silicon"] :=
Cubic[{16.57, 7.956, 6.39} *⋆ 10 ^ 10]
(*⋆ and so forth for other materials *⋆)
Definitions closer to the style of the Wolfram data repository would be:
MaterialData["Silicon", "MassDensity"] := 2332.
(*⋆ and so forth for other materials and other properties *⋆)
Then, the matrix is converted into a fourth-rank tensor by using the Voigt correspondance between indices.
MatrixToTensor[m_] := SparseArray[{
{i_, j_, k_, l_} ⧴ m[[Voigt[i, j], Voigt[k, l]]]
}, {3, 3, 3, 3}]
Finally, the Christoffel tensor is computed in the fom of a contracted tensor product: in the formula above, v is the unit vector along the
propagation direction; in the program, v is normalized.
ChristoffelTensor[c_, v_] :=
TensorContract[TensorProduct[c, Normalize[v], Normalize[v]], {{2, 5}, {4, 6}}]
The phase velocities are the square roots of the eigenvalues divided by the mass density while the associated eigenvectors give the
orthogonal polarizations, displayed below in blue; black is used to display the frame in which the elastic coefficients are assumed to be
given and the direction of propagation.
PropagationModes[material_, direction_] :=
MapAt[Sqrt[# /∕ MassDensity[material]] &,
Eigensystem[ChristoffelTensor[MatrixToTensor[
StiffnessMatrix[StiffnessCoefficients[material]]],
direction]],
1]

PropagationModes["Silicon", {1, 1, 1}]
{{9351.01, 5091.5, 5091.5}, {{-− 0.57735, -− 0.57735, -− 0.57735},
{0.806669, -− 0.293956, -− 0.512713}, {0.126299, -− 0.761746, 0.635446}}}
Graphics3D[{Map[Arrow[{{0, 0, 0}, #}] &, IdentityMatrix[3]],
{Thick, Arrow[{-− {1, 1, 1}, {1, 1, 1}} /∕ 2 /∕ Sqrt[3]]},
Prepend[Map[Arrow[{{0, 0, 0}, #}] &, -− Last[%]], Blue]},
ViewPoint → {-− 0.35, -− 1.25, 2.5}, ImageSize → 250]

For the isotropic and cubic symmetries, pure compression and pure shear waves are obtained. In all cases, the velocities can be investigated as functions of the direction of propagation. Actually, the slowness (inverse of the velocity) is represented in the form of a spherical
plot; a material is represented by 3 slowness surfaces, two of them being possibly identical in some degenerate cases. Taking into account
the symmetries, the eighth of a plot is enough.
PropagationSlowness[material_, direction_] :=
Map[Sqrt[MassDensity[material] /∕ #] &,
Eigenvalues[ChristoffelTensor[MatrixToTensor[
StiffnessMatrix[StiffnessCoefficients[material]]],
direction]],
1]
SphericalPlot3D[
Evaluate[
Chop[PropagationSlowness["Silicon", {Cos[φ] Sin[θ], Sin[φ] Sin[θ], Cos[θ]}]]],
{θ, 0, π /∕ 2}, {φ, 0, π /∕ 2}, ViewPoint → {-− 1.7, -− 2.8, 0.1}, ImageSize → 300,
Lighting → {{"Ambient", White}, {"Directional", Gray, {-− 1, 1, -− 1} *⋆ 25}}]
(*⋆ Please be patient *⋆)

Finally, the isotropic case can be tackled analytically. A purely longitudinal mode and a degenerate purely transversal mode are obtained,
as expected.
MassDensity["Isotropic"] := ρ
StiffnessCoefficients["Isotropic"] :=
Isotropic[{μ, λ}]
SimplifyPropagationModes["Isotropic", {a, b, c}],
a2 + b2 + c2 ⩵ 1, Abs[a]2 + Abs[b]2 + Abs[c]2 ⩵ 1
μ

,


ρ

μ

,

λ+2μ

c
, -−

ρ

ρ

a

, 0, 1, -−

b

a b
, 1, 0,  , , 1
a
c c

As an application, the Lamé coefficients can be derived from a measure of the velocities. The 3 modes of the general case explain why and
how an incident wave on a plane mechanical diopter generates up to 3 reflected waves and up to 3 refracted waves.

Viscoelastic models
Viscoelasticity theory is intended to model material damping in continuum mechanics [E, F]. It can be used to set up models for complex
acoustic impedances. A common practice consists in reasoning about assemblies of springs and dampers, and introduce the stiffness S and
the flexibility F as dual variables, characterized by the following relations (in the Fourier space) with σ the stress and ϵ the strain:
σ(ω) = S(ω) ϵ(ω) ⟺ ϵ(ω) = F(ω) σ(ω) with :

F(ω) = 1 /∕ S(ω)

The underlying state space relations are convolution products. The program below, which uses a cross recursivity, computes the stiffness
or the flexibility of any combination of series or parallel connections.
stiffness[Spring[k_], w_: ω] := k
stiffness[Damper[h_], w_: ω] := ⅈ w h
stiffness[InParallel[s_], w_: ω] :=
Total[Map[stiffness[#, w] &, s]]
stiffness[is : InSeries[s_], w_: ω] :=
1 /∕ flexibility[is, w]
Stiffness[in : (_InSeries _InParallel), w_: ω] :=
Together[stiffness[in, w]]
flexibility[Spring[k_], w_: ω] := 1 /∕ k
flexibility[Damper[h_], w_: ω] := -− ⅈ /∕ (w h)
flexibility[ip : InParallel[s_], w_: ω] :=
1 /∕ stiffness[ip, w]
flexibility[InSeries[s_], w_: ω] :=
Total[Map[flexibility[#, w] &, s]]
Flexibility[in : (_InSeries _InParallel), w_: ω] :=
Together[flexibility[in, w]]
Here are the classical basic models of viscoelasticity theory.
KelvinVoigt[k_, h_] := InParallel[{Spring[k], Damper[h]}]
Maxwell[k_, h_] := InSeries[{Spring[k], Damper[h]}]
Zener[c_, k_, h_] := InParallel[{Spring[c], Maxwell[k, h]}]
A variant of that notation, open to further extensions and generalizations, could be:
DipoleModel["Zener"][c_, k_, h_] :=
InParallel[{Spring[c], Maxwell[k, h]}]
Flexibility[Maxwell[k, h]]
-− ⅈ k + h ω
hkω
Stiffness[Zener[c, k, h]]
c k + ⅈ c h ω + ⅈ h k ω
k + ⅈ h ω
If need be, equivalent computations can be done in terms of Laplace transforms or even directly with the convolution product (state space
approach). Actually, it appears that this mechanical dipole approach does not really characterize the materials, for different connections of
dipoles may lead to the same stiffness or flexibility. Taking into account these are rational fractions, which testify of an underlying
differential relation, this leads to directly introduce stiffness or flexibility models in the form of rational fractions of the Laplace variable or

If need be, equivalent computations can be done in terms of Laplace transforms or even directly with the convolution product (state space
approach). Actually, it appears that this mechanical dipole approach does not really characterize the materials, for different connections of
dipoles may lead to the same stiffness or flexibility. Taking into account these are rational fractions, which testify of an underlying
differential relation, this leads to directly introduce stiffness or flexibility models in the form of rational fractions of the Laplace variable or
even differential models in terms of polynomials of the Dirac distribution and its derivatives, then supplement accordingly the previous
tranformation rules.
Stiffness[StiffnessModel[model_, s_], ω_] := model /∕. s → ⅈ ω
Stiffness[FlexibilityModel[model_, s_], ω_] := 1 /∕ model /∕. s → ⅈ ω
Flexibility[StiffnessModel[model_, s_], ω_] := 1 /∕ model /∕. s → ⅈ ω
Flexibility[FlexibilityModel[model_, s_], ω_] := model /∕. s → ⅈ ω
Stiffness[DifferentalModel[lhs_ ⩵ rhs_, t_], ω_] :=
LaplaceTransform[lhs, t, ⅈ ω] /∕ LaplaceTransform[rhs, t, ⅈ ω]
Stiffness[DifferentalModel[lhs_ ⩵ rhs_, t_], ω_] :=
LaplaceTransform[rhs, t, ⅈ ω] /∕ LaplaceTransform[lhs, t, ⅈ ω]
For instance, various expressions of the Zener model can be given that way.
StiffnessFlexibilityModel

k+hs
c k + (c + k) h s

, s, ω

c k + ⅈ h (c + k) ω
k + ⅈ h ω
Stiffness[
DifferentalModel[ k DiracDelta[t] + h DiracDelta '[t] ==
c k DiracDelta[t] + (c + k) h DiracDelta '[t], t], ω]
c k + ⅈ h (c + k) ω
k + ⅈ h ω
For thermodynamic reasons, there are nonetheless restrictions about these polynomials and their respective degrees. These stiffness or
flexibility models can then be directly adapted to impedance computations.

Results
A course in acoustic modeling gave an opportunity to initiate students into research by introducing innovating aspects. Indeed, the
coupling of symbolic manipulations with the modeling approach is still missing from the educational literature, mostly too specialized,
thus constituting a reserve of investigation topics. In practice, a combination of mathematical modeling and rule based programming leads
to quite short programs, which is an obligation in such a teaching context.

Discussion
Many subjects are based on Green functions and the Kirchhoff integral in the one-dimensional case; these lend themselves well to
symbolic manipulations, hence analytical results. In the one-dimensional case, the Kirchhoff integral boils down to values at the boundaries of an interval. It gives a formula for the field over the interval, that depends only on the boundary values; matrix (fourpole) representations derive from it.
Resorting to a symbolic language leads to a computational approach to modeling, which structures thinking because it requires rigor.
Moreover, problems are directly thought in computer language terms: data structures (and their associated data types) and functions,
procedures or transformation rules that apply to them. The computational approach also emphasizes genericity, which means a program
applies to classes of problems rather than particular ones. It also incites to tackle modeling, like programming, on the basis of specifications: the assumptions, the input data and their types, and the expected output data.
Even if most of these computations can be (and have already been) done by hand, these become faster and safer with a symbolic system
and the programs enable heavier computations in the case of more complex systems (e.g., guides with numerous sections, multilayered
materials, anisotropic materials…). Moreover, the purpose remains to give basic skills to students about quite simple problems, not to
investigate professional issues.
Because of the weak available time, the exercise is quite difficult for students, even by taking into account the assistance given in the
handout and during the workshop. Clearly, students are not used to autonomous work (which might be a typical French educational
failing), so renewing educational methods is urgent. The subjects might nonetheless be too difficult, in the sense that these require a good
level in mathematics, skills in symbolic manipulations and above all the need to combine mathematical and computational skills.

Future subjects
Because such an educational operation requires some renewal, novel subjects are in preparation, about the following topics: fluid-structure
interaction and sound radiation, surface waves and gallery modes, symbolic approximations with multipolar expansions or by the RitzGalerkin method…

Conclusion
Computer algebra and symbolic programming being far from widely spread in the scientific literature, especially in textbooks, these
constitute a means to introduce some innovating aspects in educational exercises, hence to initiate students into research.
As a side effect, this investigation stimulated more fundamental reflections about the meaning of the Kirchhoff integral or about the
emergence of an algorithmic approach to modeling and computing.
The reaction of students incites to broaden the available schedule or to introduce simpler exercises in modeling and symbolic
manipulations.

Contact

